

30 日でできる! OS 自作入門

## 30 天自制操作系统

读书笔记

【日】川合秀実 著周自恒 李黎明 曾祥江 张文旭 译

## 前言

本文档为《30天自制操作系统》(人民邮电出版社)一书的读书笔记。

文档发布在 Github 上,将随着阅读进度不定期更新。请访问 https://github.com/mengyingchina/osask-notes 获取文档最新的版本。

原书的版权声明一节,有:

本书中文简体字版由 Mynavi Corporation 授权人民邮电出版社独家出版。未经出版者书面许可,不得以任何方式复制或抄袭本书内容。

由于部分内容摘自书中,不清楚这样摘录部分内容会不会存在版权问题,如果有版权问题,我会及时删除相关内容,请知情者告知<sup>1</sup>,谢谢!

<sup>&</sup>lt;sup>1</sup>Email:mail@wanhu.me

# 目 录

第0天	着手开发之前	1
1	前言	1
2	何谓操作系统	2
3	开发操作系统的各种方法	2
4	无知则无畏	
5	如何开发操作系统	3
6	操作系统开发中的困难	3
7	学习本书时的注意事项(重要!)	3
8	各章内容摘要	3
第1天	从计算机结构到汇编程序入门	4
1	先动手操作	4
2	究竟做了些什么	
3	初次体验汇编程序	

目	习		ii
	4	加工润色	7
第	2 天	汇编语言学习与 Makefile 入门	11
	1	介绍文本编辑器	11
	2	继续开发	11
	3	先制作启动区	17
	4	Makefile 入门	17
第	3 天	进入 32 位模式并导入 C 语言	20
	1	制作真正的 IPL	20
	2	试错	23
	3	读到 18 扇区	25
	4	读入 10 个柱面	27
	5	着手开发操作系统	27
	6	从启动区执行操作系统	27
	7	确认操作系统的执行情况	27
	8	32 位模式前期准备	27
	9	开始导入 C 语言	27
	10	实现 HLT(harib00j)	27

## 第 0 天 着手开发之前

## 1 前言

阅读本书几乎不需要相关储备知识,这一点稍后还会详述。不管是用什么编程语言,只要是曾经写过简单的程序,对编程有一些感觉,就已经足够了(即使没有任何编程经验,应该也能看懂),因为这本书主要就是面向初学者的。书中虽然有很多 C 语言程序,但实际上并没有用到很高深的 C 语言知识,所以就算是曾经因为 C 语言太难而中途放弃的人也不用担心看不懂。当然,如果具备相关知识的话,理解起来会相对容易一些,不过即使没有相关知识也没关系,书中的说明都很仔细,大家可以放心。

本书以 IBM PC/AT 兼容机(也就是所谓的 Windows 个人电脑)为对象进行说明。

第 0 天 着手开发之前 2

- 2 何谓操作系统
- 3 开发操作系统的各种方法

### 4 无知则无畏

当我们打算开发操作系统时,总会有人从旁边跳出来,罗列出一大堆专业术语,问这问那,像内核怎么做啦,外壳怎么做啦,是不是单片啦,是不是微内核啦,等等。虽然有时候提这些问题也是有益的,但一上来就问这些,当然会让人无从回答。

要想给他们一个满意答复,让他们不再从旁指手画脚的话,还真得多学习,拿出点像模像样的见解才行。但我们是初学者,没有必要去学那些麻烦的东西,费时费力且不说,当我们知道现有操作系统在各方面都考虑得如此周密的时候,就会发现自己的想法太过简单而备受打击没了干劲。如果被前人的成果吓倒,只用这些现有的技术来做些拼拼凑凑的工作,岂不是太没意思了。

所以我们这次不去学习那些复杂的东西,直接着手开发。就算知道一大堆专业术语、专业理论, 又有什么意思呢?还不如动手去做,就算做出来的东西再简单,起码也是自己的成果。而且自己先实 际操作一次,通过实践找到其中的问题,再来看看是不是已经有了这些问题的解决方案,这样下来更 能深刻地理解那些复杂理论。不管怎么说,反正目前我们也无法回答那些五花八门的问题,倒不如直 接告诉在一旁指手画脚的人们:我们就是想用自己的方法做自己喜欢的事情,如果要讨论高深的问题, 就另请高明吧。

作者苦口婆心地说了这么多就是希望如果你想开发个操作系统,就动手去写吧,到底自己重写个操作系统有 什么用倒可以先放着。 第 0 天 着手开发之前 3

如果你到现在还对要不要读这本书,或者读这本书的期望的收获有疑问,推荐你阅读豆瓣上本书的一篇评 论<sup>1</sup>后再自行决定。

这本书对基础知识要求不高,懂点 C 语言和 CPU 基本知识就可以了,适合初学者。要是奔着了解操作系统原理或内核的期望,就不适宜读这本书了。30 天后也许你真的可以向作者那样做出一个基本的系统模型,但这并不意味着你对内存管理、进程管理、设备管理有着怎样高深的认识。读这本书之前先弄清自己的定位吧,毕竟时间宝贵。

- 5 如何开发操作系统
- 6 操作系统开发中的困难
- 7 学习本书时的注意事项(重要!)
- 8 各章内容摘要

 $<sup>^{1}\ \</sup>mathrm{http://book.douban.com/review/}5606888/$ 

## 第 1 天 从计算机结构到汇编程序入门

### 1 先动手操作

随书附带了光盘1,给出了书中的全部示例程序,以及部分用到的工具。

打开附带光盘,里面有一个名为 tolset 的文件夹,把这个文件夹复制到硬盘的任意一个位置上。现在里面的东西还不多,只有 3MB 左右,不过以后我们自己开发的软件也都要放到这个文件夹里,所以往后它会越来越大,因此硬盘上最好留出 100MB 左右的剩余空间。工具安装到此结束,我们既不用修改注册表,也不用设定路径参数,就这么简单。而且以后不管什么时候,都可以把这整个文件夹移动到任何其他地方。用这些工具,我们不仅可以开发操作系统,还可以开发简单的 Windows 应用程序或OSASK 应用程序等。

示例程序在附带光盘中名为 projects 的目录下,只要需要的示例程序目录复制到 tolset 文件夹里,就可以正常运行示例程序了。

考虑到开发中使用真实的软盘很不方便,作者特意准备了一个模拟器。

<sup>&</sup>lt;sup>1</sup>下载链接: http://pan.baidu.com/share/link?shareid=541099&uk=3657658273 或自行搜索"30 天自制操作系统.iso"

我们有了这个模拟器,不用软盘,也不用终止 Windows,就可以确认所开发的操作系统启动以后的动作,很方便呢。

使用模拟器的方法也非常简单,我们只需要在用!cons\_nt.bat² (或者是!cons\_9x.bat) 打开的命令行窗口中输入 "run"指令就可以了。然后一个名叫 QEMU 的非常优秀的免费 PC 模拟器就会自动运行。

§

在这一节中,作者使用二进制编辑器(十六进制编辑器)做了一个 helloos.img 文件出来。先输入了一些内容,并把内容保存成软盘映像文件格式,将这个文件写入软盘,并用它来启动电脑。画面上会显示出"hello, world"这个字符串。如果有兴趣,希望自己尝试,请参考书中这一小节的内容。

## 2 究竟做了些什么

简单解释了为什么上一节可以用二进制来写一个所谓的操作系统(虽然只能显示"hello, world"这个字符串)。

### 3 初次体验汇编程序

好,现在就让我们马上来写一个汇编程序,用它来生成一个跟刚才完全一样的 helloos.img 吧。我们这次使用的汇编语言编译器是笔者自己开发的,名为"nask",其中的很多语法都模仿了自由软件里

<sup>&</sup>lt;sup>2</sup>要根据 Windows 的版本决定用哪一个。后缀为 9x 代表是 Windows 9X 系统,后缀为 nt 的代表使用 NT 架构的 Windows 系统,如 Windows XP 及其以后的版本,以后默认为运行 cons nt.bat,并在后面将其简写为!cons。

PS: 作者开发这个系统是 2000 年左右的事情,写书的时间也比较早,所以考虑了这些问题,可以理解哈!

享有盛名的汇编器"NASM",不过在"NASM"的基础之上又提高了自动优化能力。

 $\dagger^3$  projects\01\_day\helloos1\

```
- helloos.nas
     DΒ
           0xeb, 0x4e, 0x90, 0x48, 0x45, 0x4c, 0x4c, 0x4f
           0x49, 0x50, 0x4c, 0x00, 0x02, 0x01, 0x01, 0x00
      DΒ
           0x02, 0xe0, 0x00, 0x40, 0x0b, 0xf0, 0x09, 0x00
     DΒ
     DΒ
           0x12, 0x00, 0x02, 0x00, 0x00, 0x00, 0x00, 0x00
           0x40, 0x0b, 0x00, 0x00, 0x00, 0x00, 0x29, 0xff
     DΒ
           0xff, 0xff, 0xff, 0x48, 0x45, 0x4c, 0x4c, 0x4f
     DΒ
           0x2d, 0x4f, 0x53, 0x20, 0x20, 0x20, 0x46, 0x41
     DΒ
           0x54, 0x31, 0x32, 0x20, 0x20, 0x20, 0x00, 0x00
     DB
     RESB
             16
            0xb8, 0x00, 0x00, 0x8e, 0xd0, 0xbc, 0x00, 0x7c
     DB
10
           0x8e, 0xd8, 0x8e, 0xc0, 0xbe, 0x74, 0x7c, 0x8a
     DB
11
           0x04, 0x83, 0xc6, 0x01, 0x3c, 0x00, 0x74, 0x09
     DΒ
           0xb4, 0x0e, 0xbb, 0x0f, 0x00, 0xcd, 0x10, 0xeb
     DB
13
           Oxee, Oxf4, Oxeb, Oxfd, OxOa, OxOa, Ox68, Ox65
     DB
14
           0x6c, 0x6c, 0x6f, 0x2c, 0x20, 0x77, 0x6f, 0x72
     DB
15
     DB
            0x6c, 0x64, 0x0a, 0x00, 0x00, 0x00, 0x00
     RESB
              368
```

<sup>&</sup>lt;sup>3</sup>源代码在随书光盘(见第1节的说明)中的路径; 另外,为了便于查看,给代码中添加了行号,导致复制文中代码不是很方便,请直接使用光盘中的源代码或手动输入。

```
DB 0x00, 0x00, 0x00, 0x00, 0x00, 0x00, 0x55, 0xaa
```

- DB 0xf0, 0xff, 0xff, 0x00, 0x00, 0x00, 0x00
- 20 RESB 4600
- DB 0xf0, 0xff, 0xff, 0x00, 0x00, 0x00, 0x00
- 22 RESB 1469432

把 helloos1 文件夹复制粘帖到 tolset 文件夹里,我们只要在用"!cons"打开的命令行窗口里输入"asm",就可以生成 helloos.img 文件。在用"asm"作成 img 文件后,再执行"run"指令,就可以得到与刚才一样的结果。

§

DB 指令是"data byte"的缩写,也就是往文件里直接写入1个字节的指令。

RESB 指令是 "reserve byte"的略写,如果想要从现在的地址开始空出 10 个字节来,就可以写成 RESB 10,意思是我们预约了这 10 个字节 (大家可以想象成在对号入座的火车里,预订了 10 个连号座位的情形)。而且 nask 不仅仅是把指定的地址空出来,它还会在空出来的地址上自动填入 0x00,所以我们这次用这个指令就可以输出很多的 0x00,省得我们自己去写 18 万行程序了,真是帮了个大忙。

这里还要说一下,数字的前面加上 0x,就成了十六进制数,不加 0x,就是十进制数。这一点跟 C 语言是一样的。

## 4 加工润色

刚才我们把程序变成了短短的 22 行,这成果令人欣喜。不过还有一点不足就是很难看出这些程序是干什么的,所以我们下面就来稍微改写一下,让别人也能看懂。

#### †projects\01\_day\helloos2

```
——— helloos.nas —
1; hello-os
2 ; TAB=4
4: 以下这段是标准 FAT12 格式软盘专用的代码
      DB
           0xeb, 0x4e, 0x90
           "HELLOIPL" : 启动区的名称可以是任意的字符串(8字节)
      DB
                    ;每个扇区(sector)的大小(必须为 512 字节)
           512
      DW
                    ;簇(cluster)的大小(必须为1个扇区)
      DB
           1
           1 ; FAT 的起始位置(一般从第一个扇区开始)
      DW
10
                ; FAT 的个数 (必须为 2)
      DB
           2
11
           224 : 根目录的大小(一般设成 224 项)
      DW
12
               ; 该磁盘的大小(必须是 2880 扇区)
           2880
      DW
13
               ; 磁盘的种类(必须是 0xf0)
           0xf0
      DΒ
                ; FAT 的长度(必须是 9 扇区)
      DW
15
                ; 1 个磁道(track)有几个扇区(必须是 18)
      DW
           18
           2 ; 磁头数 (必须是 2)
      DW
17
           0 ; 不使用分区, 必须是 0
      DD
           2880 : 重写一次磁盘大小
      DD
19
           0,0,0x29 ; 意义不明, 固定
      DB
```

```
; (可能是) 卷标号码
         DD
               Oxffffffff
21
               "HELLO-OS ";磁盘的名称(11 字节)
         DB
22
                            ;磁盘格式名称(8字节)
               "FAT12 "
         DB
23
                             ; 先空出 18 字节
               18
         RESB
24
25
26;程序主体
         DB
               0xb8, 0x00, 0x00, 0x8e, 0xd0, 0xbc, 0x00, 0x7c
27
               0x8e, 0xd8, 0x8e, 0xc0, 0xbe, 0x74, 0x7c, 0x8a
         DΒ
28
               0x04, 0x83, 0xc6, 0x01, 0x3c, 0x00, 0x74, 0x09
         DΒ
29
               0xb4, 0x0e, 0xbb, 0x0f, 0x00, 0xcd, 0x10, 0xeb
         DB
30
               Oxee, Oxf4, Oxeb, Oxfd
         DB
31
32
   信息显示部分
34
               0x0a, 0x0a
                            ; 2 个换行
         DB
35
         DB
               "hello, world"
36
                             ;换行
         DB
               0x0a
37
         DB
               0
38
39
               0x1fe-$
                            ; 填写 0x00, 直到 0x001fe
         RESB
40
         DB
               0x55, 0xaa
41
42
```

43; 以下是启动区以外部分的输出

```
44
45 DB Oxf0, Oxff, Oxff, Ox00, Ox00, Ox00, Ox00, Ox00
46 RESB 4600
47 DB Oxf0, Oxff, Oxff, Ox00, Ox00, Ox00, Ox00
48 RESB 1469432
```

首先是";"命令,这是个注释命令。

其次是 DB 指令的新用法。我们居然可以直接用它写字符串。在写字符串的时候,汇编语言会自动地查找字符串中每一个字符所对应的编码,然后把它们一个字节一个字节地排列起来。这个功能非常方便,也就是说,当我们想要变更输出信息的时候,就再也不用自己去查字符编码表了。

再有就是 DW 指令和 DD 指令,它们分别是"data word"和"data double-word"的缩写,是 DB 指令的"堂兄弟"。word 的本意是"单词",但在计算机汇编语言的世界里,word 指的是"16 位"的意思,也就是 2 个字节。"double-word"是"32 位"的意思,也就是 4 个字节。对了,差点忘记说 RESB 0x1fe-\$ 了。这个美元符号的意思如果不讲,恐怕谁也搞不明白,它是一个变量,可以告诉我们这一行现在的字节数(如果严格来说,有时候它还会有别的意思,关于这一点我们明天再讲)。在这个程序里,我们已经在前面输出了 132 字节,所以这里的 \$ 就是 132。因此 nask 先用 0x1fe 减去 132,得出 378 这一结果,然后连续输出 378 个字节的 0x00。

那这里我们为什么不直接写 378, 而非要用 \$ 呢? 这是因为如果将显示信息从 "hello, world"变成 "this is a pen."的话,中间要输出 0x00 的字节数也会随之变化。换句话说,我们必须保证软盘的第 510 字节 (即第 0x1fe 字节) 开始的地方是 55 AA。如果在程序里使用美元符号 (\$) 的话,汇编语言会自动计算需要输出多少个 00. 我们也就可以很轻松地改写输出信息了。

## 第 2 天 汇编语言学习与 Makefile $\lambda$ 门

### 1 介绍文本编辑器

如中文译者所注,推荐使用 Notepad++ 文本编辑器。

使用这个软件打开光盘中提供的源代码会出现日语注释乱码,选择格式 -> 编码字符集 -> 日文 -> Shift-JIS 即可正常显示代码中原书作者的日语注释。但是,关闭文件之后重新打开又会恢复原状,解决方法为在选择 Shift-JIS 编码后复制内容到一个新的文件中去,保存替换原先的文件即可,需要提示的是,新建的文件要使用 ANSI 编码格式保存,不能是 UTF-8,否则会导致后面编译时报错。

原书作者推荐的文本编辑器 TeraPad 在中文系统中会出现软件本身的文本乱码,如菜单栏工具栏的文字,但是无需设置就可以正常显示代码中的日语注释。

### 2 继续开发

选讲程序的核心部分,核心程序之前和启动区以外的内容需要具备软盘方面的相关知识,后面讲。 †projects\02 day\helloos3

```
_ helloos.nas ___
ı; hello-os
2 ; TAB=4
                 0x7c00
                               ; 指明程序的装载地址
        ORG
6; 以下的记述用于标准 FAT12 格式软盘
        JMP
                 entry
                0x90
        DB
10 --- 中略---
11;程序核心
12
13 entry:
                 AX,0 ; 初始化寄存器
        VOM
14
                 SS,AX
        VOM
15
                 SP,0x7c00
        VOM
        VOM
                 DS,AX
17
                 ES,AX
        VOM
18
19
        VOM
                 SI,msg
20
21 putloop:
```

22	MOV	AL,[SI]	
23	ADD	SI,1	; 给 SI 加 1
24	CMP	AL,O	
25	JE	fin	
26	MOV	AH,0x0e	; 显示一个文字
27	MOV	BX,15	; 指定字符颜色
28	INT	0x10	;调用显卡 BIOS
29	JMP	putloop	
30 fin:			
31	HLT		; 让 CPU 停止,等待指令
32	JMP	fin	; 无限循环
33			
34 msg:			
35	DB	0x0a, 0x0a	; 换行两次
36	DB	"hello, world"	
37	DB	0x0a	; 换行
38	DB	0	

§

ORG 指令:程序从指定的这个地址开始,也就是把程序装载到内存中的指定地址。这里是 0x7c00。

JMP 指令: 无条件跳转。配合下面的标签"entry"等,可指定跳转的目的地。

"entry":标签声明,用于指定 JMP 指令的跳转地址。

MOV 指令: 赋值。"MOV AX,0" 相当于 "AX=0" 这样一个赋值语句。同样, "MOV SS,AX" 相当于 "SS=AX"。

§

相关寄存器:

16 位寄存器: AX、CX、DX、BX、SP、BP、SI、DI:

其中, 前四个的高低 8 位可当 8 位寄存器用, AL、CL、DL、BL、AH、CH、DH、BH;

32 位寄存器: 16 位寄存器可以扩展为 32 位寄存器, EAX、ECX、EDX、EBX、ESP、EBP、ESI、EDI;

段寄存器: ES、CS、SS、DS、FS、GS;

§

"MOV SI,msg": 这里是可以"将标号赋值给寄存器"。在汇编语言中,所有的标号都仅仅是单纯的数字。每个标号对应的数字,是由汇编器根据 ORG 指令计算出来的。编译器计算出的"标号的地方对应的内存地址"就是那个标号的值。这里,msg 的地址是 0x7c74,所以这个指令就是把 0x7c74 带入到 SI 寄存器去。

"MOV AL,[SI]": MOV 指令的数据传送源和传送目的地不仅可以是寄存器或常数,也可以是内存地址。这个时候,我们使用方括号([])来表示内存地址。另外,可以用来指定内存地址的寄存器只有 BX、BP、SI、DI 这几个。

可以用下面指令将 SI 地址的 1 字节内容读入到 AL:

MOV AL, BYTE [SI]

由于 MOV 指令的规则,即源数据和目的数据必须位数相同,也就是向 AL 里代入的只能是 BYTE,这样以来就可以省略 BYTE,即可以写成:

MOV AL, [SI]

§

ADD 指令是加法指令。若以 C 语言的形式改写 "ADD SI,1"的话,就是 "SI=SI=1"。

CMP 是比较指令。"CMP AL,0",是将 AL 中的值与 0 进行比较。

JE 是条件跳转指令之一。所谓的条件跳转指令,就是根据比较的结果决定跳转或者不跳转。就 JE 指令而言,如果比较结果相等,则跳转到指定的地址;而如果比较结果不等,则不跳转,继续执行下一条指令。

CMP AL, 0

JE fin

这两条指令相当于:

if(AL == 0){ goto fin;}

§

INT 是软件中断指令。在这里是调用显卡 BIOS,不解释。

§

HLT 指令,让 CPU 停止动作,进入待机状态。

§

用 C 语言改写 helloos.nas 程序节选。

1 entry:

AX = 0;

SS = AX;

```
SP = 0x7c00;
     DS = AX;
     ES = AX;
     SI = msg;
8 putloop:
     AL = BYTE [SI];
     SI = SI+1;
     if (AL ==0) {goto fin;}
     AH = 0x0e;
12
     BX = 15;
     INT 0x10;
14
     goto putloop;
16 fin:
     HLT;
      goto fin;
```

就是有了这个程序,我们才能把 msg 里写的数据,一个字符一个字符地显示出来,并且数据变成 0 以后,HLT 指令就会让程序进入无限循环,"hello,world"就是这样显示出来的。

§

程序中的 ORG 后地址 0x7c00 是因为目前约定内存的0x00007c00-0x00007dff 地址为启动区内容的装载地址,不能随便改成其它地址。

## 3 先制作启动区

考虑到以后的开发,不要一下子用 nask 来制作整个磁盘映像,而是先用它来制作 512 字节的启动区,剩下的部分我们用磁盘映像管理工具来做。

先把 helloos.nas 的后半部分截掉,这是因为启动区只需要最后的 512 字节。现在这个程序仅仅用来制作启动区,所以把文件名改为 ipl.nas。

然后改造 asm.bat,将输出的文件名改成 ipl.bin。另外,也顺便输出列表文件 ipl.lst。这是一个文本文件,可以用来简单地确认每个指令是怎么翻译成机器语言的。

另外还增加了一个 makeimg.bat 文件。它是以 ipl.bin 为基础,制作磁盘映像文件 helloos.img 的批处理文件。利用作者开发的磁盘映像管理工具 edimg.exe,先读入一个空白的磁盘映像文件,然后在开头写入 ipl.bin,最后输出名为 helloos.img 的磁盘映像文件。

这样,从编译到测试的步骤为双击!cons,然后在命令行窗口中按顺序输入 asm ->makeimg ->run 这 3 个命令。

下一节有更简单的编译方式。

## 4 Makefile λί]

作者编写了一个 Makefile 文件,这样就可以方便的生成所需要的文件。 †projects\02\_day\helloos5

Makefile —

2# デフォルト動作

```
3
4 default :
     ../z_tools/make.exe img
7#ファイル生成規則
9 ipl.bin : ipl.nas Makefile
      ../z_tools/nask.exe ipl.nas ipl.bin ipl.lst
11
12 helloos.img : ipl.bin Makefile
      ../z_tools/edimg.exe imgin:../z_tools/fdimgOat.tek \
         wbinimg src:ipl.bin len:512 from:0 to:0 imgout:helloos.img
15
16 # コマンド
17
18 asm :
     ../z_tools/make.exe -r ipl.bin
20
21 img :
      ../z_tools/make.exe -r helloos.img
24 run :
```

```
../z tools/make.exe img
25
      copy helloos.img ..\z_tools\qemu\fdimage0.bin
26
      ../z_tools/make.exe -C ../z_tools/qemu
27
28
29 install:
      ../z_tools/make.exe img
      ../z_tools/imgtol.com w a: helloos.img
32
33 clean :
      -del ipl.bin
      -del ipl.lst
35
36
37 src only:
      ../z_tools/make.exe clean
      -del helloos.img
```

使用方法为:用!cons 打开命令行窗口,然后就可以通过输入 make img 来生成映像文件,输入 make run 来运行,等等。可以使用的参数在 Makefile 文件中写明了。另外,当执行不带参数的 make 命令时,相当于执行 make img。

## 第 3 天 进入 32 位模式并导入 C 语言

作者给开发的操作系统起名字叫 纸娃娃操作系统——haribote os。

## 1 制作真正的 IPL

制作一个可以称为真正的 IPL (启动程序装载器), 让启动区真正的开始装载程序。

§

因为磁盘最初的 512 字节是启动区,所以要装载下一个 512 字节的内容。程序是在上一天的基础上修改的,添加了以下内容:

†projects\03\_day\harib00a

_			ipl.nas 本次添加的部分
1	MOV	AX,0x0820	1P111100 11 9C44/W H4 FF/M
2	MOV	ES,AX	
3	MOV	CH,0	; シリンダ 0

4	MOV	DH,O	; ヘッド 0
5	MOV	CL,2	; セクタ 2
6			
7	MOV	AH,0x02	; AH=OxO2 : ディスク読み込み
8	MOV	AL,1	; 1 セクタ
9	MOV	BX,0	
10	MOV	DL,0x00	; A ドライブ
11	INT	0x13	; ディスク BIOS 呼び出し
12	JC	error	

INT 0x13 是调用 BIOS 的 0x13 号函数。

下面是 BIOS 13 中断的简单说明(功能有磁盘的读、写、扇区校验、寻道)

- AH=0x02 读盘/0x03 写盘/0x04 校验/0x0c 寻道
- AL= 处理连续扇区数
- CH= 柱面号 &0xff
- CL= 扇区号(0~5 位) | (柱面号 &0x300)≫2
- DH= 磁头号
- DL= 驱动器号
- ES:BX= 缓冲地址

返回值: FLAGS.CF=0, 没有错误 AH=0; FLAGS=1 有错误, AH 保存错误码

这里, JC (jump if carry) 是条件跳转指令,如果进位标志为 1,就跳转。跳转条件看调用函数的返回值FLAG.CF。

对照程序和 BIOS 函数参数说明,可以知道我们这次使用的是读盘,柱面号是 0,磁头号是 0,扇区号是 2,磁盘号是 0。

§




图 3.1: 软盘的结构

一张软盘有 80 个柱面,2 个磁头,18 个扇区,且一个扇区有 512 字节。所以一张软盘的容量是  $80 \times 2 \times 18 \times 512 = 1474560 Byte = 1440 KB$ 。

含有 IPL 的启动区,位于 C0-H0-S1 (柱面 0, 磁头 0, 扇区 1),下一个扇区是 C0-H0-S2, 这次我们要装载的扇区就是这个。

ES:BX= 缓冲地址 是个内存地址,表明要把软盘上读出的数据装载到内存的哪个位置。由于一个 BX 只能表示  $0\sim0$ xfff 的值,即 64K,太小了,使用段寄存器以 ES:BX 这种方式来表示地址,写成"MOV AL,[ES:BX]",代表"ES×16+BX"的内存地址。程序中指定了 ES=0x0820,BX=0,所以软盘的数据会被装载到 0x8200 $\sim0$ x83ff 的位置。

§

作者使用变量的方式改写了 Makefile 文件,可以看一下。

## 2 试错

鉴于软盘的不可靠性,有时候需要在软盘读不出来的时候多读几次,这里重读 5 次。 †projects\03\_day\harib00b

```
ipl.nas 本次添加的部分

ipl.nas 本次添加的部分

MOV AX,0x0820

MOV ES,AX

MOV CH,0 ; シリンダ 0

MOV DH,0 ; ヘッド 0

MOV CL,2 ; セクタ 2
```

8			
9	MOV	SI,0	; 失敗回数を数えるレジスタ
10 retry:			
11	MOV	AH,0x02	; AH=0x02 : ディスク読み込み
12	MOV	AL,1	; 1 セクタ
13	MOV	BX,0	
14	MOV	DL,0x00	; A ドライブ
15	INT	0x13	; ディスク BIOS 呼び出し
16	JNC	fin	; エラーがおきなければ fin へ
17	ADD	SI,1	; SI に 1 を足す
18	CMP	SI,5	; SI と 5 を比較
19	JAE	error	; SI >= 5 だったら error へ
20	MOV	AH,0x00	
21	MOV	DL,0x00	; A ドライブ
22	INT	0x13	; ドライブのリセット
23	JMP	retry	

其中, JNC (jump if not carry), 进位标志为 0 的话跳转; JAE (jump if above or equal), 大于或等于时跳转。

在重新读盘之前,做了如下处理,AH=0x00,DL=0x00,INT=0x13,即完成系统复位。

## 3 读到 18 扇区

往后多读几个扇区,读完柱面 0 的 18 个扇区。 †projects\03\_day\harib00c

			ipl.nas 本次添加的部分
ı; j	ディスクを読む	3	
2			
3	MOV	AX,0x0820	
4	MOV	ES,AX	
5	MOV	CH,O	; シリンダ 0
6	MOV	DH,O	; ヘッド 0
7	MOV	CL,2	; セクタ 2
8 readloop:			
9	MOV	SI,0	; 失敗回数を数えるレジスタ
10 ret	ry:		
11	MOV	AH,0x02	; AH=OxO2 : ディスク読み込み
12	MOV	AL,1	; 1 <i>セクタ</i>
13	MOV	BX,O	
14	MOV	DL,0x00	; A ドライブ
15	INT	0x13	; ディスク BIOS 呼び出し
16	JNC	next	; エラーがおきなければ next へ
17	ADD	SI,1	; SI に 1 を足す

18	CMP	SI,5	; SI と 5 を比較
19	JAE	error	; SI >= 5 だったら error へ
20	MOV	AH,0x00	
21	MOV	DL,0x00	; A ドライブ
22	INT	0x13	; ドライブのリセット
23	JMP	retry	
24 next:			
25	MOV	AX,ES	; アドレスを 0x200 進める
26	ADD	AX,0x0020	
27	MOV	ES,AX	; ADD ES,0x020 という命令がないのでこうしている
28	ADD	CL,1	; CL に 1 を足す
29	CMP	CL,18	; CL と 18 を比較
30	JBE	readloop	; CL <= 18 だったら readloop へ

JBE(jump if below or equal), 小于等于则跳转。

要读入下一个扇区只需给 CL 加 1, 给 ES 加上 0x20 (512/16)。CL 是扇区号, ES 指定读入地址。

这里使用循环的方式读入各个扇区,而不是在开始的时候设置读入的扇区数 AL=17,因为磁盘 BIOS 读盘函数有一些"补充说明":

指定处理的扇区数,范围在 0x01 0xff (指定 0x02 以上的数值时,要特别注意能够连续处理多个扇区的条件。如果是 FD 的话,似乎不能跨越多个磁道,也不能超过 64KB 的界限。)

经过上面这些读盘处理,已经把磁盘上 C0-H0-S2 到 C0-H0-S18 的  $512 \times 17 = 8704$  个字节的内容,装载到了内存的  $0x8200\sim0xa3ff$ 。

- 4 读入 10 个柱面
- 5 着手开发操作系统
- 6 从启动区执行操作系统
- 7 确认操作系统的执行情况
- 8 32 位模式前期准备
- 9 开始导入 C 语言
- 10 实现 HLT(harib00j)